ISTITUTO NAZIONALE DI FISICA NUCLEARE
Laboratori Nazionali di Frascati

LNF-79/4(R)
11 Gennaio 1979

G. Giordano and E. Poldi Alai :
GEOMETRY OF GAUSSIAN BEAMS AND LASER CAVITIES.

Servizio Documentazione
dei Laboratori Nazionali di Frascati

INFN - Laboratori Nazionali di Frascati
Servizio Documentazione

LNF-79/4(R)
11 Gennaio 1979

G. Giordano and E. Poldi Alai: GEOMETRY OF GAUSSIAN BEAMS AND LASER CAVITIES.

1. - INTRODUCTION.

Aim of the present paper is to give a summary of the problems involving the propagation of laser beams, the dimensioning of laser cavities and optical transport systems.

In the Ladon project an optical system 17.5 m long leads the beam from the laser to the interaction point with the electrons in the middle of the Adone straight section[1,2].

Since the laser beam dimension has to be known with great accuracy in the overlapping region, the exit beam profile has to be carefully measured and its evolution through the whole optical transport system exactly computed.

Two methods for the beam profile measurements are described in Sec. 3 after a summary of the beam propagation formulae is reviewed in Sec. 1. The knowledge of the computations methods on the cavity dimensioning (Sec. 4) is the base line for any future development of the Ladon facility.

2. - PROPAGATION OF GAUSSIAN BEAMS.

The intensity distribution of a laser beam as a function of the distance r from the beam centre, is a gaussian curve, usually written as:

$$I(r) = I_o \exp(-2r^2/w^2) . \tag{1}$$

We define as "beam radius" the distance $r = w$ at which $I(w) = I_o 1/e^2$. The points of minimum radius and with a plane wavefront are called "waists".

2.1. - Propagation formulae.

The beam radius w and the wavefront curvature radius R characterize the beam in any point and are univocally determined by the beam waist w_o and the distance z from it:

$$w(z) = w_o \sqrt{1 + (z/a_o)^2} , \tag{2}$$

$$R(z) = (z^2 + a_o^2)/z , \tag{3}$$

where:

$$a_o = \pi w_o^2/\lambda \qquad (\lambda = \text{optical wavelength}) . \tag{4}$$

The beam angular divergence $\vartheta(z)$, defined as $w(z)/R(z)$ is :

$$\vartheta(z) = \frac{\lambda}{\pi w_0} \frac{1}{\sqrt{1 + (a_0/z)^2}} \quad . \tag{5}$$

The formulae (2), (3) and (5) are plotted in Fig. 1 and Fig. 2.



FIG. 1 - Radius (w) and curvature radius (R) evolutions for a gaussian beam of waist w_o as a function of the normalized distance z/a_o ($a_o = \pi w_o^2/\lambda$) from w_o. R is normalized to its minimum value $2a_o$ and w to w_o.



FIG. 2 - Behaviour of the angular divergence ϑ versus the normalized distance from the waist w_o.

As it can be seen, the angular divergence tends asimptotically at the value:

$$\vartheta_{\infty} = \lambda / \pi w_o \ . \tag{6}$$

For $z = a_o$, the curvature radius assumes its minimum value $R(a_o) = R_{min} = 2a_o$.

The point $z = a_o$ has another particularity: the value of w_o that produces at a given distance z' the minimum value $w(z')$, is just such that $z' = a_o$.

2.2. - Inverse formulae.

Sometimes one has to determine the waist dimension and position knowing the beam geometry in one or more points. By inverting formulae (2) and (3), the waist w_o and its distance z from P can be expressed in function of the beam radius w and its curvature radius R in P(3):

$$w_o = \frac{w}{\sqrt{1 + \left(\frac{\pi w^2}{\lambda R}\right)^2}} \ , \tag{7}$$

$$z = \frac{R}{1 + \left(\frac{\lambda R}{\pi w^2}\right)^2} \ . \tag{8}$$

If one knows the beam radii w_1 and w_2 in two points separated by a distance d, w_o and z (distance between w_o and w_1) are:

$$w_o = \sqrt{\frac{B \pm \sqrt{B^2 - A}}{A}} \ , \tag{9}$$

$$z = \pm \frac{\pi w_o}{\lambda} \sqrt{w_1^2 - w_2^2} \, , \qquad (10)$$

where :

$$A = (\frac{\pi}{d\lambda})^4 (w_2^2 - w_1^2)^2 + (\frac{2\pi}{d\lambda})^2 \, ,$$

$$B = (\frac{\pi}{d\lambda})^2 (w_1^2 + w_2^2) \, .$$

2. 3. - Lenses.

The beam wavefront transformation due to a spherical lens with focal length F is given by[4]:

$$\frac{1}{R_+} = \frac{1}{R_-} - \frac{1}{F} \, , \qquad (11)$$

where R_+ and R_- are the wavefront curvature radii after and before the lens. In our conventions the sign of R_+, R_- and F are positive for diverging beams and convergent lenses. A spherical mirror with curvature radius M, is equivalent to a lens with focal length $F = M/2$.

The waists w_1 and w_2 before and after a lens F and their distances d_1 and d_2 from the lens are related by the formulae[4]:

$$w_2 = \left[\frac{1}{w_1^2} (1 - \frac{d_1}{F})^2 + (\frac{\pi w_1}{F\lambda})^2 \right]^{-1/2} \, , \qquad (12)$$

$$d_2 = F + \frac{(d_1 - F) F^2}{(d_1 - F)^2 + (\frac{\pi w_1^2}{\lambda})^2} \, , \qquad (13)$$

where $d_1 > 0$ for an incoming diverging beam and $d_2 > 0$ for an outgoing converging beam (see Fig. 3).

Then, using only the formulae (12) and (13) it is possible to compute the evolution of a laser beam crossing any system of lenses and mirrors. In Appendix A we illustrate a computer program written for this aim.

3. - MEASUREMENT OF LASER BEAM RADIUS.

As we said in the previous section, the knowledge of the beam radii in two points along the beam path is sufficient to determine the dimension and position of the waist and then the beam geometry in any point.

We have used two methods to measure beam radii, both with the same principle that is to cut the beam with a moveable dark screen and measure the light power as a function of the screen position. The first method is the following:

A dark screen is moved through the beam spot by a micrometer slit with steps usually of 1/10 mm. For each position we read the "transmitted" power and then we make a plot of the power difference from a point to the following one. This difference is proportional to the average va-



FIG. 3 - The six possible cases of beam transformation by a lens.

lue of the beam intensity distribution in the interval between the two points. Infact, this difference can be written as (this is true only for the TEM_{00}) (see Fig. 4) :

$$P(x_o) - P(x_o + \Delta x) \propto \int_{-\infty}^{+\infty} dy \exp(-2y^2/w^2) \int_{-\infty}^{x_o} dx \exp(-2x^2/w^2) -$$

$$- \int_{-\infty}^{+\infty} dy \exp(-2y^2/w^2) \int_{-\infty}^{x_o+\Delta x} dx \exp(-2x^2/w^2) = \qquad (14)$$

$$= k \int_{x_o}^{x_o+\Delta x} dx \exp(-2x^2/w^2) \simeq k\, \Delta x \exp(-2x_o^2/w^2) ,$$

(where : $k = \int_{-\infty}^{+\infty} dy \exp(-2y^2/w^2)$).



FIG. 4 - Set up of the first method for measurement of beam radius. The slit moves through the beam in the x direction by step Δx.

Then the plots of these differencies is a 'step' reproduction of the beam intensity distribution (Fig. 5), and from it we can determine the beam radius. Usually we make an half dozen measures in different points along the beam path with the slit moving both in the horizontal and the vertical direction. In this way it is possible to test the presence of transverse modes.

We have recently accomplished a faster and more automatic measurement method in the following way :

The moveable dark screen is repleaced by an helix (Fig. 6) rotating with a fixed and stable period (in our case it was about 1 sec). The distance between the helix centre and the beam spot center is about 16 cm, such that $2\pi R = 1$ m. The time behaviour of the light power read after the helix is just connected, through the linear velocity of the helix cutting edge, to the x integral of the



FIG. 5 - Reproduction of the beam intensity distribution obtained by the slit method (see text). Solid line is a fit by eyes.



FIG. 6 - Set up of the second method for measurements of the beam radius. The helix rotates with constant angular velocity ω cutting the beam spot.

beam intensity distribution. Really, in this method, the cut line does not procede parallel to itself, causing an error due to the difference between ϑ and $\sin\vartheta$, ϑ being of the order of the ratio of the beam diameter (for example 1 mm) to the helix radius (16 cm).

From the beam intensity integral, seen on a scope through a fast photodiode set after the helix, it is already possible to determine the beam radius by measuring the integral risetime (Fig. 7a). Infact, it is easy to see that, in the case of a purely gaussian beam, the risetime t_r is :

$$t_r = 1.28 \frac{w}{\omega R} , \qquad (15)$$

where : ω = helix angular velocity.



(a) (b)

FIG. 7 - Reproduction of the beam intensity distribution obtained by the helix method (see text). Fig. 7a is the distribution integral (direct photodiode response). Fig. 7b is the intensity distribution (derived photodiode response).

As we said in the introduction, we need to measure the profile of the laser beam used in the Ladon project that now is produced by an SP 171 Argon-Ion laser with cavity-dumper. In Fig. 8 is



FIG. 8 - Cavity-dumper set up. M_1, M_2 and M_3 are high reflection mirrors, PT is a total reflection prism, LT the laser tube, P the prism for wavelength selection. The modulator is an acoustooptical deflector working in the Bragg regime[(2)].

sketched the laser apparatus setup and in Fig. 9 are reported the results of the beam profile measurements. As it can be seen from the comparison between the experimental points and the theoric curve obtained by a standard fit program, the accuracy of the measures is about 3%.



FIG. 9 - Results of the cavity dumper beam profile measurement. Solid line is the best fit of the experimental points.

4. - GEOMETRY OF LASER CAVITIES.

Let's now examine the problem of computing the dimensions of the beam inside a resonant cavity.

The most simple and used configuration of a laser cavity consists in a plane and a concave mirror. At the end mirrors, the beam must have the same curvature radius of the mirror; then it has a waist on the plane mirror. Inverting (3), with R(z) equal to the concave mirror curvature radius R and z equal to the cavity length d, one can determine w_o, resulting:

$$w_o^4 = \left(\frac{\lambda}{\pi}\right)^2 d\,(R - d)\,. \tag{16}$$

In a similar way, one can compute the waist size and position when the cavity is composed by two curved mirrors. Let R_1 and R_2 be two concave mirrors and d_1 and d_2 the distance between them and the waist w_o ($d_1 + d_2$ = d the cavity length); then, w_o, d_1 and d_2 are given by:

$$w_o^4 = \frac{\left(\frac{\lambda}{\pi}\right)^2 d(R_1 - d)(R_2 - d)(R_1 + R_2 - d)}{(R_1 + R_2 - 2d)^2}\,, \tag{17}$$

$$d_1 = \frac{d(R_2 - d)}{(R_1 + R_2 - 2d)}\,, \qquad d_2 = \frac{d(R_1 - d)}{(R_1 + R_2 - 2d)}\,. \tag{18}$$

In the case of a cavity composed by a concave mirror (R_1) and a convex mirror (R_2), the beam profile inside the cavity is obtained by considering a waist w_o set outside the cavity at a distance z from R_2, with :

$$w_o^4 = \left(\frac{\lambda}{\pi}\right)^2 \frac{d(R_1 - d)(-R_2 - d)(R_1 - R_2 - d)}{(R_1 - R_2 - 2d)^2} , \qquad (19)$$

$$z = - \frac{d(R_1 - d)}{(R_1 - R_2 - 2d)} . \qquad (20)$$

Up to now, we have considered cavities composed only of the two end mirrors. The problem of finding the beam profile in a cavity containing a lens, or more than a lens, complicates a bit and requires an approach different from the pure application of the beam propagation formulae, that is the transfer matrix method.

Each optical element (distances, lenses, mirrors) of the cavity is described by a 2 x 2 matrix (Fig. 10). These matrices give the beam evolution through the optical element, with the beam



FIG. 10 - Matrix representation of some optical elements.

beeing described in any point along the cavity by a complex parameter q defined as :

$$\frac{1}{q} = \frac{1}{R} - i \frac{\lambda}{\pi w^2} , \qquad (21)$$

where λ is the optical wavelength, w the beam radius and R its curvature radius. The q parameters before (q_1) and after (q_2) an optical element described by the matrix $\begin{pmatrix} A & B \\ C & D \end{pmatrix}$ are linked by the so called "ABCD law"(4) :

$$q_2 = \frac{Aq_1 + B}{Cq_1 + D} . \qquad (22)$$

It can be easily verified that (22) is equivalent to (2), (3) and (11).

Let's define as total transfer matrix of a cavity the matrix obtained as the product of all the transfer matrices associated to the optical elements the beam passes in a complete double transit. Considering now the cavity shown in Fig. 11, the total matrix TM calculated from the point P going to the left is :

$$TM = \begin{pmatrix} 1 & d_4 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ -\frac{2}{R_2} & 1 \end{pmatrix} \begin{pmatrix} 1 & d_2 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ -\frac{1}{F} & 1 \end{pmatrix} \begin{pmatrix} 1 & d_1 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ -\frac{2}{R_1} & 1 \end{pmatrix} \begin{pmatrix} 1 & d_1 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ -\frac{1}{F} & 1 \end{pmatrix} \begin{pmatrix} 1 & d_3 \\ 0 & 1 \end{pmatrix} . \quad (23)$$



FIG. 11 - Example of a cavity with an internal lens.

Let's write it as :

$$TM = \begin{pmatrix} A & B \\ C & D \end{pmatrix} = \begin{pmatrix} 1 & d_4 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} A' & B' \\ C' & D' \end{pmatrix} \begin{pmatrix} 1 & d_3 \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} A' + C'd_4 & B' + A'd_3 + D'd_4 + C'd_3d_4 \\ C' & D' + C'd_3 \end{pmatrix} . \quad (24)$$

It has been demonstrated by (3) that the condition of stability for the cavity is :

$$-1 < \frac{A + D}{2} < +1 \ . \quad (25)$$

$N = (A + D)/2$ is called "stability number". The value of N is independent of the way (starting point and versus) that one choice to compute the total matrix.

In parti cular, when the cavity is only composed by two mirrors R_1 and R_2 spaced by d, the stability condition can be written as :

$$0 < (1 - d/R_1)(1 - d/R_2) < 1 \ . \quad (26)$$

If (25) is satisfied, the beam starting from P with a given value of q, will return to P with the same q, that is :

$$q = \frac{Aq + B}{Cq + D} \ , \quad (27)$$

that gives :

$$\frac{1}{q} = \frac{D - A}{2B} \pm i \frac{\sqrt{4 - (A + D)^2}}{2B} \ . \quad (28)$$

From the definition of q it follows that, for having a waist in P it must be :

$$D = A \ . \quad (29)$$

From (24) one has :

$$A' + C'd_3 = D' + C'd_4 = D' + C'(d_2 - d_3) \quad , \tag{30}$$

and then one finds the waist position d_3 :

$$d_3 = (C'd_2 + D' - A')/2C' \quad . \tag{31}$$

The waist w_o is found through (21) and (27) to be :

$$w_o^2 = \frac{\lambda}{\pi} \frac{2B}{\sqrt{4 - (A+D)^2}} \quad . \tag{32}$$

Once one knows the position and value of the waist in an interval of the cavity, it is possible to compute it in all the other intervals by (12) and (13), or repeating the same procedure for each interval. The latter method is used in the computer program reported in Appendix B.

REFERENCES.

(1) - G. Matone et al., A monochromatic and polarized photon beam for photonuclear reactions. The Ladon project at Frascati, Photonuclear Reactions (Erice, 1976), vol. 2.
(2) - G. Giordano and G. Matone, Laser light modulation: the acoustooptical effect and the cavity dumping technique, Frascati Report LNF-77/32 (1977).
(3) - H. Kogelnik and T. Li, Laser beam and resenators, Applied Optics 5, 1550 (1966).
(4) - H. Kogelnik, Imaging of optical modes-resonators with internal lenses, Bell System Techn. Journal 44, 455 (1965).

APPENDIX A

```
C       PROPAGATION OF A LASER BEAM THROUGH AN OPTICAL SYSTEM
C       COMPOSED WITH N LENSES OF FOCAL LENGTHS F(I)
C       SPACED BY THE DISTANCES DIST(I).
C       THE BEAM STARTS FROM A WAIST W(1) AT DIST(1) FROM F(1).
C       WL(I) ARE THE BEAM RADII ON THE LENSES. D(I) ARE THE DISTANCES
C       BETWEEN THE WAISTS W(I) AND THE LENSES F(I).
        DIMENSION F(4),D(4),DIST(4),W(4),WL(4)
        N=3
        M=N+1
        WL(M)=0.
        PG=3.1415927
        WRITE(6,400)
        READ(5,500) XL
        WRITE(6,600)
        READ(5,500) W(1)
        READ(5,100) DIST
        READ(5,100) F
        D(1)=DIST(1)
        DO 1 K=1,N
        X=PG*W(K)/XL
        T=1.-(D(K)/F(K))
        W(K+1)=((T/W(K))**2)+((X/F(K))**2)
        W(K+1)=1./SQRT(W(K+1))
        Z=(T**2)+((X*W(K)/F(K))**2)
        Z=F(K)*(1.-(T/Z))
        D(K+1)=DIST(K+1)-Z
        IF(K.EQ.N) D(K+1)=-D(K+1)
        WL(K)=1.+(D(K)/(X*W(K)))**2
1       WL(K)=W(K)*SQRT(WL(K))
        WRITE(6,200)
        DO 2 I=1,M
2       WRITE(6,300) I,W(I),DIST(I),D(I),F(I),WL(I)
        STOP
100     FORMAT(8F10.5)
200     FORMAT(/,4X,'I',10X,'W',14X,'DIST',15X,'D',16X,'F',15X,'WL')
300     FORMAT(4X,I1,5(5X,E12.5))
400     FORMAT($,2X,'LAMBDA =')
500     FORMAT(E12.5)
600     FORMAT($,2X,'WAIST =')
        END
```



The inputs of the program are :

F(i)	are the focal lengths of the N lenses composing the system (F(N + 1) = 0).
DIST(i)	are the distances between the starting waist w_o and the first lens, and between the other lenses (DIST (N + 1) = 0).
LAMBDA	is the optical wavelength.
WAIST	is the starting waist w_o.

The outputs of the program are :

D(i)	are the distances between the waist and the following lens. The last D(i) is the distance between the last waist and lens.
W(i)	are the waists.
WL(i)	are the beam radii on the lenses (WL (N + 1) = 0).

APPENDIX B

```
C
C       COMPUTATION OF THE GEOMETRY OF A LASER CAVITY
C
        DIMENSION EL(5),RIS(9),T(9,2,2),TT(10,2,2)
        DIMENSION U(2,2),PM(2,2),TM(2,2)
        N=5
        MDT=2*N-1
        MDTT=2*N
        MDEL=N
        MDRIS=MDT
        U(1,1)=1.
        U(1,2)=0.
        U(2,1)=0.
        U(2,2)=1.
        READ(5,303) XLANDA
        READ(5,100) EL
        DO 21 K=1,MDRIS
   21   RIS(K)=0.
        DO 4 K=1,MDT
        CALL JUVE(T,K,MDT,U,1,1,2,2)
        IF(K.GT.1) GO TO 4
        CALL JUVE(TT,1,MDTT,U,1,1,2,2)
    4   CONTINUE
        JLM=((N-3)/2)+1
        JAK=0
        DO 5 JL=1,JLM
        JEL=JL-1
        IF(JEL.NE.0) GO TO 8
        CALL DRAKEN(EL,N,T,MDT)
C
C       COMPUTE THE TOTAL TRANSFER MATRIX
C
        CALL JUVE(TT,2,MDTT,T,1,MDT,2,2)
        KMAX=2*N-3
        KMIN=2
        CALL HAWK(MDT,MDTT,T,TT,KMIN,KMAX)
        KMAX=KMAX+1
        CALL JUVE(PM,1,1,TT,KMAX,MDTT,2,2)
        GO TO 6
    8   KMAX=3+(2*(JEL-1))
        CALL HAWK(MDT,MDTT,T,TT,1,KMAX)
        KMIN=KMAX+2
        CALL JUVE(TT,KMIN,MDTT,U,1,1,2,2)
        CALL HAWK(MDT,MDTT,T,TT,KMIN,MDT)
        KMIN=KMIN-1
        KMAX=MDTT
        CALL MIRAGE(TT,KMIN,MDTT,TT,KMAX,MDTT,U,1,1,PM)
        M=(2*N)-2
        CALL JUVE(TT,M,MDTT,PM,1,1,2,2)
    6   CONTINUE
        JJ=MDTT-(4*(JEL+1))
        JH=2*(JEL+1)
C
C       BEGIN COMPUTING BEAM DIMENSIONS
C
        RIS(JJ)=PM(2,1)*EL(JH)
        RIS(JJ)=RIS(JJ)+PM(2,2)-PM(1,1)
        RIS(JJ)=RIS(JJ)/(2.*PM(2,1))
        RIS(JJ+2)=EL(JH)-RIS(JJ)
        IF((RIS(JJ).GT.-0.1).AND.(RIS(JJ).LT.0.1)) RIS(JJ)=0.
```

```
      IF((RIS(JJ+2).GT.-0.1).AND.(RIS(JJ+2).LT.0.1)) RIS(JJ+2)=0.
      IF((RIS(JJ).GE.-0.1).AND.(RIS(JJ+2).GE.-0.1)) GO TO 9
      WRITE(6,900) JH
      IF(JEL.NE.0) GO TO 5
      T(MDT,1,2)=EL(JH)
      GO TO 5
    9 T(2*N-2,1,2)=RIS(JJ)
      T(2*N-1,1,2)=RIS(JJ+2)
      KMIN=MDTT-2
      KMAX=MDT
      CALL MIRAGE(T,KMIN,MDT,TT,KMIN,MDTT,T,KMAX,MDT,TM)
C
C     END COMPUTING OF THE TOTAL TRANSFER MATRIX
C
      IF(JEL.NE.0) GO TO 7
      X7=T(2*N-2,1,2)
      X8=T(2*N-1,1,2)
    7 T(2*N-2,1,2)=X7
      T(2*N-1,1,2)=X8
C
C     COMPUTING THE STABILITY NUMBER
C
      STAB=(TM(1,1)+TM(2,2))/2.
      IF(JAK.EQ.0) WRITE(6,600) STAB
      JAK=JAK+1
      IF((STAB.LE.-1.).OR.(STAB.GE.1.)) GO TO 11
C
C     END COMPUTING STABILITY
C
      X=TM(1,1)+TM(2,2)
      X=SQRT(4.-X*X)
      X=2.*TM(1,2)/X
      X=XLANDA*X/3.1415927
      X=ABS(X)
      RIS(JJ+1)=SQRT(X)
      IF(JEL.EQ.0) GO TO 5
      X=RIS(JJ)
      RIS(JJ)=RIS(JJ+2)
      RIS(JJ+2)=X
C
C     END COMPUTING FIRST BEAM DIMENSIONS
C
    5 CONTINUE
      KMAX=(N-3)/2+1

C
C     COMPUTING LAST BEAM DIMENSIONS
C
      DO 22 KK=1,KMAX
      K=KK
      K=MDTT-(4*K)+1
      RIS(K+2)=PW(MDRIS,K,K+1,RIS,XLANDA)
   22 RIS(K-2)=PW(MDRIS,K,K-1,RIS,XLANDA)
C
C     END COMPUTING BEAM DIMENSIONS
C
      DO 10 K=1,MDRIS
   10 WRITE(6,800) K,RIS(K)
   11 CONTINUE
   99 STOP
  100 FORMAT(5E12.5)
  200 FORMAT(4F10.5)
  300 FORMAT(/,50X,'F=',F8.3,20X,'R =',F8.3)
  303 FORMAT(E12.5)
  600 FORMAT(/,5X,'STABILITY NUMBER =',F10.5)
  800 FORMAT(5X,'RIS(',I2,' ) =',E15.8)
  900 FORMAT(/,40X,'IN EL(',I2,') THERE IS NO WAIST')
      END
```

```
      SUBROUTINE JUVE(T1,L,MDT1,T2,M,MDT2,IX,JX)
      DIMENSION T1(MDT1,IX,JX),T2(MDT2,IX,JX)
C     TRANSFERS IN T1(L) THE VALUES OF T2(M)
      DO 1 I=1,IX
      DO 1 J=1,JX
    1 T1(L,I,J)=T2(M,I,J)
      RETURN
      END
```

```
      SUBROUTINE HAWK(MDT,MDTT,T,TT,KMIN,KMAX)
C     COMPUTES T(KMAX)*T(KMAX-1)*...*T(KMIN)*TT(KMIN)
C     EXIT  TT(KMAX+1)
      DIMENSION T(MDT,2,2),TT(MDTT,2,2)
      DO 1 K=KMIN,KMAX
      DO 1 I=1,2
      DO 1 J=1,2
      X=0.
      DO 2 L=1,2
    2 X=X+T(K,I,L)*TT(K,L,J)
    1 TT(K+1,I,J)=X
      RETURN
      END
```

```
      SUBROUTINE DRAKEN(EL,N,T,MDT)
C     COMPUTES THE TRANSFER MATRICES FOR EL(K)
      DIMENSION EL(N),T(MDT,2,2)
      KMAX=2*N-3
      DO 1 KK=1,KMAX
      K=KK
      IF(K.GT.N) GO TO 3
      M=MOD(K,2)
      IF(M.EQ.0) GO TO 2
      T(K,2,1)=0.
      IF(EL(K).NE.0.) T(K,2,1)=-1./EL(K)
      GO TO 1
    2 T(K,1,2)=EL(K)
      GO TO 1
    3 M=2*N-K
      CALL JUVE(T,K,MDT,T,M,MDT,2,2)
    1 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MIRAGE(T1,M,MDT1,T2,MM,MDT2,T3,MMM,MDT3,EXM)
C     COMPUTES EXM(I,J) = T1(M)*T2(MM)*T3(MMM)
      DIMENSION T1(MDT1,2,2),T2(MDT2,2,2),T3(MDT3,2,2),EXM(2,2)
      DO 1 K=1,2
      DO 1 I=1,2
      DO 1 J=1,2
      X=0.
      DO 2 L=1,2
      IF(K.EQ.2) GO TO 3
      X=X+T1(M,I,L)*T2(MM,L,J)
      GO TO 2
    3 X=X+EXM(I,L)*T3(MMM,L,J)
    2 CONTINUE
    1 EXM(I,J)=X
      RETURN
      END
```

```
      FUNCTION PW(MDRIS,N,M,RIS,XL)
C     COMPUTES BEAM RADIUS AT DISTANCE RIS(M) FROM THE WAIST RIS(N)
      DIMENSION RIS(MDRIS)
      PW=RIS(N)
      IF((RIS(M).EQ.0.).OR.(RIS(N).EQ.0.)) GO TO 30
      PW=RIS(N)**2*3.1415927
      PW=RIS(M)*XL/PW
      PW=SQRT(1.+PW**2)*RIS(N)
30    RETURN
      END
```



The inputs of the program are :

XLANDA is the optical wavelength.

EL(i) are the optical elements composing the cavity : focal lengths(for the mirrors write R/2) and distances. The sign of the focal lengths is positive for converging lenses and concave mirrors. N is their number (it must be odd) (See Figure).

The outputs of the program are :

RIS(i) are the beam radii on the lenses and mirrors, the waists and their distances from the previous and the following lens or mirror (See Figure).